

## 4 Einführung in Maple

### 4.1 Grundlagen

#### 4.1.1 Was ist Maple?

Maple ist ein kommerzielles Softwarepaket für das symbolische Rechnen; man spricht auch von einem Computeralgebra-System. Es kann z.B. algebraische Ausdrücke bearbeiten, allgemeine Lösungen von Gleichungssystemen finden, Grenzwerte von Funktionen bestimmen, Integrale auswerten oder (partielle) Ableitungen von Funktionen ausrechnen. In diesem Sinne erspart uns Maple das Nachschlagen in Formelsammlungen.

Maple rechnet mit ganzen Zahlen (bzw. rationalen Zahlen) exakt, indem es bei Bedarf mit beliebig vielen Stellen arbeitet. Dies kann allerdings zu Speicherplatz- und Rechenzeitproblemen führen.


Maple besitzt auch eine gute Graphik; darauf wird hier jedoch nicht eingegangen (da wir für Graphiken Matlab verwenden).

#### 4.1.2 Aufruf von Maple

Unter Linux startet man Maple durch Eingabe des Kommandos

```
xmple &
```

und erhält dann ein neues Fenster:



Das Erscheinungsbild des Maple-Fensters hängt von der aktuellen Version ab, welche in der obersten Zeile (hier Maple 11) erkennbar ist.

In diesem Fenster (weiße Fläche) werden die Maple-Befehle eingegeben. Dabei können Sie mit

`View` → `Zoom Factor`

die Schriftgröße ändern.

Am linken Rand erkennen Sie einige Zusatz-Tools, z.B. `Greek` für die Eingabe von griechischen Buchstaben oder `Expressions` für die Eingabe von Integralen, Summen oder Produkten.

### 4.1.3 Arbeitsspeicher

Die während einer Sitzung eingegebenen Kommandos werden in einen Arbeitsspeicher (maple worksheet) geschrieben. Dieser kann vor dem Verlassen in eine Datei gespeichert werden. Dazu klicken Sie

`File` → `Save As...`

an und geben dann den Dateinamen `<name>` ein. Diese erhält den Zusatz `.mw` (`mw` = maple worksheet). Daran erkennt dann Maple, dass darin ein Maple-Sitzung gespeichert wurde und in einer späteren Maple-Sitzung durch

`File` → `Open...`

wieder geladen werden kann. Innerhalb einer Maple-Sitzung wird der Arbeitsspeicher durch

`restart`

gelöscht.

### 4.1.4 Maple beenden

Man beendet Maple durch Anklicken von

`File` → `Exit`

### 4.1.5 Online-Hilfe

Eine Online-Hilfe erhält man durch Anklicken von `help` in der Kopfzeile. In dem dann erscheinenden Menü können verschiedene Punkte ausgewählt werden. Für den Anfänger interessant ist vor allem

`Maple Help` .

Dadurch erscheint ein Inhaltsverzeichnis, und Sie können dann die entsprechenden Themen anklicken.

Informationen über einen bestimmten Befehl bekommen Sie durch Eingabe des Befehls mit vorangestelltem `?`, zum Beispiel

`?diff` .

### 4.1.6 Eingabe von Maple-Kommandos

Maple arbeitet (ebenso wie Matlab) interaktiv, d.h. jeder Maple-Befehl wird nach Betätigen der Eingabetaste sofort ausgeführt.

Dabei ist zu beachten:

- Es wird zwischen Groß- und Kleinbuchstaben unterschieden.
- Das Zeichen `#` dient als Kommentarzeichen: Der Text rechts davon bis zum Zeilenende wird als Kommentar betrachtet (und somit nicht als Maple-Befehl interpretiert).
- Ein Maple-Kommando kann sich über mehrere Zeilen erstrecken, muss dann jedoch am Zeilenende mit dem Zeichen `\` markiert werden. Alternativ kann auch `Shift + Enter` verwendet werden.
- Ein Maple-Kommando kann mit einem Doppelpunkt abgeschlossen werden. Dann wird das Ergebnis dieses Befehls nicht im Maple-Fenster angezeigt.
- In einer Zeile dürfen mehrere Maple-Befehle stehen. Diese müssen jedoch mit Strichpunkt (Ergebnis wird angezeigt) oder Doppelpunkt (Ergebnis wird nicht angezeigt) getrennt werden.
- `?` vor einem Befehl aktiviert die Online-Hilfe zu diesem Kommando.
- `%` bezieht sich auf den zuvor berechneten Ausdruck.

Maple rechnet ganzzahlig und rational exakt (beliebig viele Stellen).

### Beispiele:

```
> m := 5;
                                     m := 5
> p := m*m;
                                     p := 25
> q := 2^p;
                                     q := 33554432
> r := 5^(p^2);
r := 7182120874830735080661624773480024744364634020656043207139670780774366417110744230376\
7251899701806274179648528361693395908493607789072590515205376005217038518864343357974\
3241720259632199259695945742908504868334649410427176109294374402530861274131588811191\
1087856525487071480712730550125031653424356470048912459287045495641072957730169031770\
8800960633255792684347901301624401393481617554653310899100138176009977541980333626270\
294189453125
> s := r/(7^36);
s := 7182120874830735080661624773480024744364634020656043207139670780774366417110744230376\
7251899701806274179648528361693395908493607789072590515205376005217038518864343357974\
3241720259632199259695945742908504868334649410427176109294374402530861274131588811191\
1087856525487071480712730550125031653424356470048912459287045495641072957730169031770\
8800960633255792684347901301624401393481617554653310899100138176009977541980333626270\
294189453125/2651730845859653471779023381601
> |
```

Beachten Sie, dass in Maple `:=` als Wertzuweisungszeichen dient. `s` ist eine rationale Zahl. Suchen Sie den Bruchstrich!

Die Zahlen können auch in der Gleitpunktdarstellung ausgegeben werden:

```
evalf(s)
```

bzw.

```
Digits:=40
evalf(s)
```

liefert `s` als Gleitpunktzahl mit 10 Stellen (Voreinstellung) bzw. mit 40 Stellen.

#### 4.1.7 Konstanten und Standardfunktionen

Maple kennt alle wichtigen Konstanten und Standardfunktionen. Wir geben hier nur eine kleine Auswahl an.

##### Konstanten

<code>I</code>	imaginäre Einheit ( $I^2 = -1$ )
<code>Pi, pi</code>	Kreiszahl $\pi$ . Für (numerische) Berechnungen muss <code>Pi</code> verwendet werden.
	<code>evalf(Pi)</code> $\rightsquigarrow$ 3.141592654
	<code>evalf(pi)</code> $\rightsquigarrow$ $\pi$
<code>infinity</code>	$\infty$
<code>true</code>	logische Größe <i>wahr</i>
<code>false</code>	logische Größe <i>falsch</i>

##### Standardfunktionen

<code>cos(x)</code>	Kosinus
<code>sin(x)</code>	Sinus
<code>exp(x)</code>	Exponentialfunktion
<code>ln(x)</code>	natürlicher Logarithmus
<code>log10(x)</code>	Zehnerlogarithmus
<code>log[b](x)</code>	Logarithmus zur Basis $b$
<code>log[2](x)</code>	Zweierlogarithmus
<code>sqrt(x)</code>	Quadratwurzel
<code>root[n](x), root(x,n)</code>	$n$ -te Wurzel
<code>binomial(n,k)</code>	Binomialkoeffizienten $\binom{n}{k}$
<code>n!</code>	Fakultäten

Ausführliche Informationen über die Standardfunktionen erhalten Sie über die Onlinehilfe:

```
Help → Maple Help → Mathematics → Basic Mathematics
      → Initially Known Functions
```

#### 4.1.8 Maple Programme

Man kann mehrere Befehle in eine Textdatei schreiben, deren Name frei wählbar ist (zum Beispiel `programm1.txt`), und dann in Maple ausführen durch das Kommando

```
read 'programm1.txt'
```

Achten Sie auf den richtigen Akzent. Die einzelnen Kommandos müssen dabei mit Se-

mikolon (Ergebnis erscheint auf dem Bildschirm) bzw. Doppelpunkt (keine Ausgabe auf den Bildschirm) abgeschlossen werden. Die in dieser Datei enthaltenen Maple-Befehle werden dann sequentiell abgearbeitet.

**Beispiel:** Die Datei `programm1.txt` habe folgenden Inhalt:

```
# Maple-Programm
x := Pi/4;
y := cos(x);
z := cos(x)^2 + sin(x)^2;
v := (x + y)/z;
evalf(v);
```

## 4.2 Symbolisches Rechnen

### 4.2.1 Ausdrücke bearbeiten

Maple bietet folgende Möglichkeiten:

<code>simplify(&lt;ausdruck&gt;)</code>	Ausdruck vereinfachen
<code>expand(&lt;ausdruck&gt;)</code>	Ausdruck ausrechnen

**Beispiele:**

```
z := (5/3)^n * 3^n * (x^2 + 2*x + 1) / (x + 1)
```

$$\frac{\left(\frac{5}{3}\right)^n 3^n (x^2 + 2x + 1)}{x + 1}$$

```
simplify(z)
```

$$(x + 1)5^n$$

```
expand((x+1)^4*(x+2)^2)
```

$$x^6 + 8x^5 + 26x^4 + 44x^3 + 41x^2 + 20x + 4$$

### 4.2.2 Faktorisierungen

Für die Faktorisierung (Primzahlzerlegung) von ganzen Zahlen gibt es das Kommando

```
ifactor(<zahl>)
```

zum Beispiel

```
ifactor(97518330)
```

$$(2) (3)^7 (5) (7)^3 (13)$$

Für die Faktorisierung von Polynomen bietet Maple die Kommandos

```
factor(<ausdruck>),
factor(<ausdruck>, <ueber>)
```

Sind die Koeffizienten des Polynoms ganzzahlig, so wird über  $\mathbb{Z}$  faktorisiert.

**Beispiele:**

```
factor(4*x^4-28*x^3+52*x^2+12*x-72)
```

$$4(x-3)^2(x+1)(x-2)$$

```
factor(x^3+3)
```

$$x^3 + 3$$

```
factor(x^3+3,3^(1/3))
```

$$(x^2 - x 3^{(1/3)} + 3^{(2/3)})(x + 3^{(1/3)})$$

```
factor(x^3+3,real))
```

$$(x + 1.442249570)(x^2 - 1.442249570 x + 2.080083822)$$

```
factor(x^3+3,complex)
```

$$(x+1.442249570)(x+(-0.7211247852+1.249024766I))(x+(-0.7211247852-1.249024766I))$$

**4.2.3 Summen, Reihen und Produkte**

Maple kennt dafür die Kommandos

```
sum(<ausdruck> , <bereich> ) ,
product(<ausdruck> , <bereich> ) .
```

**Beispiele:**

$$\text{sum}(1/k, k=1..20) \quad \left[ \text{berechnet} \quad \sum_{k=1}^{20} \frac{1}{k} \right]$$

$$\frac{55835135}{15519504}$$

$$\text{sum}(1/(k!), k=0..infinity) \quad \left[ \text{berechnet} \quad \sum_{k=0}^{\infty} \frac{1}{k!} = e \right]$$

$$\text{sum}(k^2, k=1..n) \quad \left[ \text{berechnet} \quad \sum_{k=1}^n k^2 \right]$$

$$\frac{1}{3}(n+1)^3 - \frac{1}{2}(n+1)^2 + \frac{1}{6}n + \frac{1}{6}$$

```
simplify(%)
```

$$\frac{1}{3}n^3 + \frac{1}{2}n^2 + \frac{1}{6}n$$

$$\text{product}(2^k, k=1..8) \quad \left[ \text{berechnet} \quad \prod_{k=1}^8 2^k \right]$$

$$68719476736$$

Selbstverständlich dürfen diese Kommandos auch geschachtelt werden:

$$\text{sum}(\text{product}(1/i^2, i=1..k), k=2..6) \quad \left[ \text{berechnet} \sum_{k=2}^6 \prod_{i=1}^k \frac{1}{i^2} \right]$$

$$\frac{144937}{518400}$$

#### 4.2.4 Grenzwerte

Die Berechnung von Grenzwerten erfolgt mit dem Befehl

$$\text{limit}(\langle \text{ausdruck} \rangle, \langle \text{stelle} \rangle)$$

**Beispiele:**

$$\text{limit}(\sin(x)/x, x=0) \quad \left[ \text{berechnet} \lim_{x \rightarrow 0} \frac{\sin(x)}{x} \right]$$

$$1$$

$$\text{limit}(\exp(-x^2), x=\text{infinity}) \quad \left[ \text{berechnet} \lim_{x \rightarrow \infty} \exp(-x^2) \right]$$

$$0$$

#### 4.2.5 Umgang mit Funktionen

Funktionen werden in Maple wie folgt definiert:

$$f := x \rightarrow \sin(x)/x$$

$$f := x \rightarrow \frac{\sin(x)}{x}$$

$$g := (x, y) \rightarrow \text{sqrt}(x^2 + 3*y^4)$$

$$g := (x, y) \rightarrow \sqrt{x^2 + 3y^4}$$

Beim Funktionsaufruf haben wir zwei Möglichkeiten: *symbolisch* oder *numerisch*.

$$f(2) \quad \rightsquigarrow \quad \frac{1}{2} \sin(2)$$

$$f(2.0) \quad \rightsquigarrow \quad 0.4546487134$$

$$\text{evalf}(f(2)) \quad \rightsquigarrow \quad 0.4546487134$$

$$g(2, 1/2) \quad \rightsquigarrow \quad \frac{1}{4} \sqrt{67}$$

$$g(2.0, 0.5) \quad \rightsquigarrow \quad 2.046338193$$

Auch eine Verkettung (Komposition) von Funktionen ist möglich, zum Beispiel

$$(\sin @ \exp)(x) \quad \rightsquigarrow \quad \sin(\exp(x))$$

$$(\ln @@ 2)(x) \quad \rightsquigarrow \quad \ln(\ln(x))$$

## 4.2.6 Ableitungen

Für die Berechnung von (partiellen) Ableitungen bietet Maple zwei Möglichkeiten. In den folgenden Beispielen seien  $f$  und  $g$  die in obigem Abschnitt definierten Funktionen.

1. Verwendung des `diff` - Kommandos.

`diff(f(x),x)` [ berechnet  $f'(x)$  ]

$$\frac{\cos(x)}{x} - \frac{\sin(x)}{x^2}$$

`diff(f(x),x,x)` [ berechnet  $f^{(2)}(x)$  ]

$$-\frac{\sin(x)}{x} - \frac{2 \cos(x)}{x^2} + \frac{2 \sin(x)}{x^3}$$

`diff(f(x),x$5)` [ berechnet  $f^{(5)}(x)$  ]

`diff(exp(-x^2),x)`

$$-2x e^{-x^2}$$

`diff(g(x,y),x)` [ berechnet  $g_x(x,y)$  ]

$$\frac{x}{\sqrt{x^2 + 3y^4}}$$

`diff(g(x,y),y,x)` [ berechnet  $g_{yx}(x,y)$  ]

Es gibt auch ein `Diff` - Kommando, das den zu differenzierenden Ausdruck hinschreibt:

`Diff(sin(x),x) = diff(sin(x),x)`

$$\frac{d}{dx} \sin(x) = \cos(x)$$

`Diff(sqrt(x^2 + y^2),x) = diff(sqrt(x^2 + y^2),x)`

$$\frac{\partial}{\partial x} (\sqrt{x^2 + y^2}) = \frac{x}{\sqrt{x^2 + y^2}}$$

2. Verwendung des Differentialoperators `D`

`D(f)` [ berechnet  $f'(x)$  ]

`D(f)(1)` [ berechnet  $f'(1)$  ]

`(D@@3)(f)` [ berechnet  $f^{(3)}(x)$  ]

`D(x->exp(x^2))`

$$x \rightarrow 2x e^{x^2}$$



`D[1](g)`      [ berechnet  $g_x(x, y)$  ]  
`D[2](g)`      [ berechnet  $g_y(x, y)$  ]  
`D[1,2](g)`      [ berechnet  $g_{xy}(x, y)$  ]  
`D[1](g)(3,4)`      [ berechnet  $g_x(3, 4)$  ]

## 4.2.7 Integrale

### 1. Unbestimmte Integrale (Stammfunktionen)

`int(exp(-t), t)`      [ berechnet Stammfunktion zu  $e^{-t}$  ]  
 $-e^{(-t)}$

`f := x -> sin(x)*cos(x)`  
`int(f(x), x)`  
 $\frac{1}{2} \sin(x)^2$

### 2. Bestimmte Integrale

`int(exp(-t), t=0..2)`      [ berechnet  $\int_0^2 e^{-t} dt$  ]  
 $-e^{-2} + 1$

`int(exp(-t), t=0.0..2.0)`  
 0.8646647168

`int(exp(-x^2), x=-infinity..infinity)`      [ berechnet  $\int_{-\infty}^{\infty} e^{-x^2} dx$  ]  
 $\sqrt{\pi}$

`Int(exp(-x^2), x=-infinity..infinity)`  
 $\int_{-\infty}^{\infty} e^{-x^2} dx$

Beachten Sie den Unterschied zwischen `int` und `Int`.

## 4.2.8 Taylor-Polynome

Die Taylor-Entwicklung einer (hinreichend oft differenzierbaren) Funktion erhält man durch

`taylor(<ausdruck>, <stelle>, <ordnung>)` .

Dabei enthält `<stelle>` den Entwicklungspunkt und die natürliche Zahl `<ordnung>` die Ordnung des Restglieds, d.h. der Grad des Taylor-Polynoms ist `<ordnung>-1`.

**Beispiele:**

```
taylor(exp(x),x=0,4)
```

$$1 + x + \frac{1}{2}x^2 + \frac{1}{6}x^3 + O(x^4)$$

```
taylor(exp(x),x=pi,3)
```

$$e^\pi + e^\pi(x - \pi) + \frac{1}{2}e^\pi(x - \pi)^2 + O((x - \pi)^3)$$

Das Kommando

```
convert(% ,polynom)
```

gibt den zuletzt berechneten Ausdruck als Polynom aus:

$$e^\pi + e^\pi(x - \pi) + \frac{1}{2}e^\pi(x - \pi)^2$$

Der Befehl

```
convert(<ausdruck> , <form>)
```

stellt einen Ausdruck in der gewünschten Form dar.

**Beispiele:**

```
convert(2011,binary)
```

```
11111011011
```

```
convert(2013,hexadecimal)
```

```
7DD
```

```
convert(101101,decimal,binary)
```

```
45
```

```
convert(AFFE,decimal,hex)
```

```
45054
```

```
convert(1.2345,fraction)
```

```
 $\frac{2469}{2000}$ 
```

**4.2.9 Gleichungen**

Zur Lösung von Gleichungssystemen gibt es das Kommando

```
solve(<gleichungen> , <variablen>)
```

**Beispiele:**

1. Die Lösungen von  $4x^4 + 8x^2 = 10$  erhält man durch

```
pol := 4*x^4 + 8*x^2 = 10
solve(pol,x)
```

Möglich ist auch

```
solve(4*x^4 + 8*x^2 = 10,x)
```

$$-\frac{1}{2}\sqrt{-4+2\sqrt{14}}, \frac{1}{2}\sqrt{-4+2\sqrt{14}}, -\frac{1}{2}\sqrt{-4-2\sqrt{14}}, \frac{1}{2}\sqrt{-4-2\sqrt{14}}$$

```
evalf(%)
```

$$-.9331820260, 0.9331820260, -1.694351998I, 1.694351998I$$

2. Die Lösungen des Gleichungssystems

$$\begin{aligned}x^2 + y^2 &= 4 \\(x - 1)^4 + (y + 1)^2 &= 10\end{aligned}$$

erhalten wir durch folgende Maple-Sequenz:

```
e1 := x^2 + y^2 = 4
e2 := (x-1)^4 + (y+1)^2 = 10
solve({e1,e2},{x,y})
evalf(%)
```

$$\{y = 2, x = 0\}, \{x = 2.928323177 + 0.1903946038 I, y = 0.259774844 - 2.146231461 I\}$$

#### 4.2.10 Mengen

Mengen werden durch die üblichen Mengenklammern  $\{ \}$  dargestellt:

```
A := {rot, blau, gelb}
B := {grün, braun, grau, gelb}
```

Vereinigung  $A \cup B$ , Durchschnitt  $A \cap B$  und Differenz  $A \setminus B$  erhalten wir durch

```
A union B
```

$$\{grau, braun, grün, blau, rot, gelb\}$$

```
A intersect B
```

$$\{gelb\}$$

```
A minus B
```

$$\{rot, blau\}$$

Möglich ist auch die Abfrage, ob ein Element zur Menge gehört. Das Ergebnis ist *true* oder *false*.

```
farbe := rot
member(farbe,A)
```

*true*

## 4.3 Vektoren und Matrizen

### 4.3.1 Vektoren

Für die Definition von Vektoren stellt Maple zwei Möglichkeiten zur Verfügung.

1. Möglichkeit:

`v := <1,2,3>` (Spaltenvektor)

$$\begin{bmatrix} 1 \\ 2 \\ 3 \end{bmatrix}$$

`w := <1|2|3>` (Zeilenvektor)

$$\begin{bmatrix} 1 & 2 & 3 \end{bmatrix}$$

2. Möglichkeit: Verwendung des Befehls `Vector`

`v := Vector([1,2,3])`

$$v := \begin{bmatrix} 1 \\ 2 \\ 3 \end{bmatrix}$$

`w := Vector[row]([1,2,3])`

$$w := \begin{bmatrix} 1 & 2 & 3 \end{bmatrix}$$

`Vector(3)`

$$\begin{bmatrix} 0 \\ 0 \\ 0 \end{bmatrix}$$

`Vector[row](1..5,2)`

$$\begin{bmatrix} 2 & 2 & 2 & 2 & 2 \end{bmatrix}$$

### 4.3.2 Matrizen

Für Matrizen haben wir ebenfalls zwei Optionen.

1. Möglichkeit:

`A := <<1|2|3>,<4|5|6>,<7|8|9>>`

$$\begin{bmatrix} 1 & 2 & 3 \\ 4 & 5 & 6 \\ 7 & 8 & 9 \end{bmatrix}$$

2. Möglichkeit: Verwendung des Befehls `Matrix`

`A := Matrix([[1,2,3],[4,5,6],[7,8,9]])`

$$\begin{bmatrix} 1 & 2 & 3 \\ 4 & 5 & 6 \\ 7 & 8 & 9 \end{bmatrix}$$

```

B := Matrix(2,3)
      [ 0 0 0 ]
      [ 0 0 0 ]

C := Matrix(1..2,1..3,2)
      [ 2 2 2 ]
      [ 2 2 2 ]

```

Der Zugriff auf die einzelnen Elemente bzw. auf Teilbereiche der Matrix erfolgt durch

```

u := A[2,3]
A[1..2,2..3] := Matrix([[1,1],[1,1]])

```

### 4.3.3 Umgang mit Matrizen

Für den Umgang mit Matrizen (Multiplikation, Eigenwerte, Determinanten,...) braucht man ein Zusatzpaket, welches durch das Kommando

```
with(LinearAlgebra)
```

bereitgestellt wird. Nach diesem Aufruf erhalten Sie eine Liste mit den nun zusätzlich verfügbaren Befehlen.

#### Beispiele:

<code>MatrixVectorMultiply(A,v)</code>	$Av$ (Matrix mal Spaltenvektor)
<code>VectorMatrixMultiply(w,A)</code>	$wA$ (Zeilenvektor mal Matrix)
<code>MatrixMatrixMultiply(A,B)</code>	$AB$ (Matrixprodukt)
<code>Eigenvalues(A)</code>	Eigenwerte von $A$
<code>Eigenvectors(A)</code>	Eigenwerte und Eigenvektoren von $A$
<code>Determinant(A)</code>	Determinante von $A$
<code>Transpose(A)</code>	Transponierte $A^T$
<code>MatrixInverse(A)</code>	inverse Matrix $A^{-1}$

Zum Schluss noch ein kleines Maple-Programm. Testen Sie es mal.

```

# Maple-Programm
with(LinearAlgebra);
A := <<1|2|3>, <0|4|5>, <0|0|6>>;
v := <1,2,3>;
c := MatrixVectorMultiply(A,v);
Eigenwerte := Eigenvalues(A);
Determinante := Determinant(A);
Transponierte := Transpose(A);
Inverse := MatrixInverse(A);

```